

Nu Game Engine

The world's first practical functional game engine!

What's It All About?

The Nu Game Engine is a **Mature, Functional, 2d (and soon 3D!) Game Engine** written in **F#**.

Let me explain each of those terms –

Mature

Nu is mature and exposes multiple levels of programmability depending on your performance needs. At the high level, there's the Elm-style programming API. At the low level, there's an Entity-Component-System API called **ECS**. By default you'll be using the Elm-style API, but you can drop down to the ECS when you need to scale to hundreds of thousands of entities

Additionally, there is a tile map system that utilizes **Tiled#**, and there is a physics system that utilizes **Farseer Physics**. Rendering, audio, and other IO systems are handled in a cross-platform way with **SDL2 / SDL2#**. There is also an asset management system to make sure your game can run on memory-constrained devices such as the iPhone. There is a declarative special effects system called, appropriately enough, **EffectSystem**, as well as an efficient **ParticleSystem**. On top of all that, there is a built-in game editor called **Gaia**!

Functional

Nu can be configured to run in an immutable mode. This is how the world editor, Gaia, implements Undo and Redo – by taking snapshots of the immutable world state. However, for optimal performance, Nu is configured to run with mutation under the hood outside of the editor.

2d Game Engine

Nu is not a code library. It is a **game software framework**. Thus, it sets up a specific way of approaching and thinking about the design of 2d* games. Nu is intended to be a broadly generic toolkit for 2d game development.

** 3D rendering and physics are currently being implemented for Nu with a projected ship date of 8/1/22, so thank you for your patience! In fact, the engine's coordinate system is already converted to 3D so that very few API changes will be incurred.*

F#

We know what F# is, so why use it? First, because of its **cross-platform** nature. Theoretically, Nu runs on both the .NET Framework and Windows, and with a bit of project tinkering, on Mono and Linux. I'm hoping to soon port it to .NET Core. But more on why F#. F# is probably the best mainstream language available for writing a cross-platform functional game engine. Unlike Clojure, F#'s **static type system** makes the code easier to reason about and dare I say more efficient. Unlike Scala, F# offers a simple and easy-to-use programming model. Unlike Haskell, you get intuitive and a well-tooled debugging experience. Unlike JVM languages generally, F# allows us to **code and debug with Visual Studio**. Finally, I speculate that game developers have more familiarity with **the .NET ecosystem** than the JVM, so that leverage is at hand.

Getting Started

Nu is made available from a **GitHub repository** located at <https://github.com/bryanedds/Nu>. To obtain it, first **fork** the repository's latest **release** to your own GitHub account (register as a new GitHub user if you don't already have an account). Second, **clone** the forked repository to your local machine (instructions here <https://help.github.com/articles/fork-a-repo>). The Nu Game Engine is now yours!

Note that unlike code libraries that are distributed via NuGet, forking and cloning the FP Works repository at GitHub is how you attain Nu. You will be happy with this if you need to make changes to the engine or step debug into it!

The next thing you must do is to install the VC 2012 redistributable **vc redistrib_x64** (link is here – <https://www.microsoft.com/en-in/download/details.aspx?id=30679>). **UPDATE: I'm no longer sure if this is necessary! YMMV without it!**

Upon inspecting your clone of the repository, the first thing you might notice about it is that the repository contains more than just the Nu Game Engine. It also includes a **Projects** folder containing the sample games, **Nelmish** (a simple GUI programming example), **Elmario** (a simple physics game example), **BlazeVector** (a somewhat more sophisticated shooter game), and my WIP role-playing game **OmniBlade** (an actively developed commercial title). You should explore these projects in the order I've listed them here.

To open the Nu solution, first make sure to have **Visual Studio 2022** installed (the free edition is fine). Then open the **Nu.sln** file in the root folder. Attempt to build the whole solution. If there is a problem with building it, try to figure it out, and failing that, ask me questions directly via bryanedds@gmail.com.

Once the solution builds successfully, ensure that the **Nelmish** project is set as the **StartUp** project, and then run the game by pressing the **|> Start** button in Visual Studio.


Creating your own Nu game Project

Next, let's build your own game project using the Nu Game Engine.

First, set the **Nu** project at the **StartUp** project, then run it by pressing the **|> Start** button in Visual Studio.

When the program runs, it will ask you if you would like to make a new game project and what you would like the name of the project to be. Once you answer these questions, it will create the desired project in the **./Projects** folder in a sub-folder of the given project name.


Next, right-click on the **Projects** solution folder in Visual Studio, then click **Add** then **Existing Project...** and select the newly-created **.fsproj** file like so -



Once you follow these simple steps, click the **OK** button to create the project. Now you can build and run the new project by setting it as the **StartUp** project and then pressing the **|> Run** button.

For those having difficulty creating a new project (such as on non-Windows platforms), you might instead just want to use the **UserProject** for your game. You can alter its code however you like, but do be aware this is a workaround and you may have to deal with merges as it is kept up to date. Windows users, try to use the previous project creation method instead of **UserProject**.

When the new project is run from Visual Studio, you'll get the basic template game that includes a splash screen, a title screen, a credits screen, and a little Mario-like gameplay screen -



Basic Nu Start-up Code

Here's the start-up code presented with comments in **Program.fs** -

```
namespace MyGame
open System
open Nu
module Program =

    // this the entry point for your Nu application
    let [<EntryPoint; STAThread>] main _ =

        // this specifies the window configuration used to display the game
        let sdlWindowConfig = { SdlWindowConfig.defaultConfig with WindowTitle = "MyGame" }

        // this specifies the configuration of the game engine's use of SDL
        let sdlConfig = { SdlConfig.defaultConfig with ViewConfig = NewWindow sdlWindowConfig }

        // use the default world config with the above SDL config
        let worldConfig = { WorldConfig.defaultConfig with SdlConfig = sdlConfig }

        // initialize Nu
        Nu.init worldConfig.NuConfig

        // run the engine with the given config and plugin
        World.run worldConfig (MyPlugin ())
```

All this code initializes Nu and instantiates the game engine.

Let's take a some code that is more interesting in **MyGame.fs** -

```
namespace MyGame
open Prime
open Nu
open Nu.Declarative

[<AutoOpen>]
module MyGame =

    // this is our Elm-style model type. It determines what state the game is in. To learn about the
    // Elm-style in Nu, see here - https://vsyncronicity.com/2020/03/01/a-game-engine-in-the-elm-style/
    type Model =
        | Splash
        | Title
        | Credits
        | Gameplay of Gameplay

    // this is our Elm-style message type. It provides a signal to show the various screens.
    type Message =
        | ShowTitle
        | ShowCredits
        | ShowGameplay

    // this is our Elm-style command type. Commands are used instead of messages when explicitly
    // updating the world is involved.
    type Command =
        | Exit

    // this extends the Game API to expose the above model as well as the model bimapped to Gameplay,
    type Game with
        member this.GetModel world = this.GetModelGeneric<Model> world
        member this.SetModel value world = this.SetModelGeneric<Model> value world
        member this.Model = this.ModelGeneric<Model> ()
        member this.Gameplay = Lens.bimap (function Gameplay gameplay -> Some gameplay | _ -> None) (constant Gameplay)

    // this is the game dispatcher that is customized for our game. In here, we create screens as
    // content and bind them up with events and properties.
    type MyGameDispatcher () =
        inherit GameDispatcher<Model, Message, Command> (Splash)

    // here we channel from gui events to signals
    override this.Channel (_, _) =
        [Simulants.Splash.Screen.DeselectEvent => msg ShowTitle
         Simulants.Title.Gui.Credits.ClickEvent => msg ShowCredits
         Simulants.Title.Gui.Play.ClickEvent => msg ShowGameplay
         Simulants.Title.Gui.Exit.ClickEvent => cmd Exit
         Simulants.Credits.Gui.Back.ClickEvent => msg ShowTitle]

    // here we link the game and gameplay models (two-way bind), then we bind the desired
    // screen based on the state of the game (or None if splashing),
    override this.Initializers (model, game) =
        [game.Gameplay <=> Simulants.Gameplay.Screen.Gameplay
         game.DesiredScreen <== model --> fun model ->
             match model with
             | Splash -> Desired Simulants.Title.Screen
             | Title -> Desired Simulants.Title.Screen
             | Credits -> Desired Simulants.Credits.Screen
             | Gameplay gameplay ->
                 match gameplay with
                 | Playing -> Desired Simulants.Gameplay.Screen
                 | Quitting -> Desired Simulants.Title.Screen]

    // here we handle the above messages
    override this.Message (_, message, _, _) =
        match message with
        | ShowTitle -> just Title
        | ShowCredits -> just Credits
        | ShowGameplay -> just (Gameplay Playing)
```

```

// here we handle the above commands
override this.Command (_, command, _, world) =
    match command with
    | Exit -> just (World.exit world)

// here we describe the content of the game, including all of its screens.
override this.Content (_, _) =
    [Content.screen Simulants.Splash.Screen.Name (Nu.Splash (Constants.Dissolve.Default, Constants.Splash.Default, None, Simulants.Title.Screen)) [] []
    Content.screenFromGroupFile Simulants.Title.Screen.Name (Dissolve (Constants.Dissolve.Default, None)) "Assets/Gui/Title.nugroup"
    Content.screenFromGroupFile Simulants.Credits.Screen.Name (Dissolve (Constants.Dissolve.Default, None)) "Assets/Gui/Credits.nugroup"
    Content.screen<MyGameplayDispatcher> Simulants.Gameplay.Screen.Name (Dissolve (Constants.Dissolve.Default, None)) [] []]

```

As mentioned in the comments, the best way to understand this code is to first read my article here -

<https://vsyncronicity.com/2020/03/01/a-game-engine-in-the-elm-style/>

It explains how you can program Nu simulants in the Elm-style, which is how the above code is programmed.

Once you understand how Nu is programmed in the Elm-style, let's take a look at how we actually define the gameplay portion of the game in **MyGameplay.fs** -

```

namespace MyGame
open Prime
open Nu
open Nu.Declarative

[<AutoOpen>]
module MyGameplay =

    // this is our Elm-style model type. Either we're playing or we're quitting back to the title screen.
    type Gameplay =
        | Playing
        | Quitting

    // this is our Elm-style message type.
    type GameplayMessage =
        | Quit

    // this is our Elm-style command type. Commands are used instead of messages when things like physics are involved.
    type GameplayCommand =
        | Jump
        | MoveLeft
        | MoveRight
        | UpdateEye
        | Nop

    // this extends the Screen API to expose the above model.
    type Screen with
        member this.GetGameplay world = this.GetModelGeneric<Gameplay> world
        member this.SetGameplay value world = this.SetModelGeneric<Gameplay> value world
        member this.Gameplay = this.ModelGeneric<Gameplay> ()

    // this is the screen dispatcher that defines the screen where gameplay takes place
    type MyGameplayDispatcher () =
        inherit ScreenDispatcher<Gameplay, GameplayMessage, GameplayCommand> (Quitting)

    // here we channel from events to signals
    override this.Channel (_, _) =
        [Simulants.Game.KeyboardKeyDownEvent => fun evt ->
            if evt.Data.KeyboardKey = KeyboardKey.Up && not evt.Data.Repeated then cmd Jump
            else cmd Nop
        Simulants.Gameplay.Screen.UpdateEvent => fun _ ->
            if KeyboardState.isKeyDown KeyboardKey.Left then cmd MoveLeft
            elif KeyboardState.isKeyDown KeyboardKey.Right then cmd MoveRight
            else cmd Nop
        Simulants.Gameplay.Screen.PostUpdateEvent => cmd UpdateEye]

```

```

// here we handle the above messages
override this.Message (_, message, _, _) =
    match message with
    | Quit -> just Quitting

// here we handle the above commands
override this.Command (_, command, _, world) =
    let world =
        match command with
        | Jump ->
            let physicsId = Simulants.Gameplay.Scene.Player.GetPhysicsId world
            if World.isBodyOnGround physicsId world then
                let world = World.applyBodyForce (v3 0.0f 140000.0f 0.0f) physicsId world
                World.playSound Constants.Audio.SoundVolumeDefault (asset "Gameplay" "Jump") world
            else world
        | MoveLeft ->
            let physicsId = Simulants.Gameplay.Scene.Player.GetPhysicsId world
            if World.isBodyOnGround physicsId world
            then World.applyBodyForce (v3 -2000.0f 0.0f 0.0f) physicsId world
            else World.applyBodyForce (v3 -500.0f 0.0f 0.0f) physicsId world
        | MoveRight ->
            let physicsId = Simulants.Gameplay.Scene.Player.GetPhysicsId world
            if World.isBodyOnGround physicsId world
            then World.applyBodyForce (v3 2000.0f 0.0f 0.0f) physicsId world
            else World.applyBodyForce (v3 500.0f 0.0f 0.0f) physicsId world
        | UpdateEye ->
            if World.getUpdateRate world <> 0L
            then Simulants.Game.SetEyeCenter (Simulants.Gameplay.Scene.Player.GetCenter world) world
            else world
        | Nop -> world
    just world

// here we describe the content of the game including the level, the hud, and the player
override this.Content (_, screen) =

    [// the gui group
    Content.group Simulants.Gameplay.Gui.Group.Name []
    [Content.button Simulants.Gameplay.Gui.Quit.Name
    [Entity.Text == "Quit"
    Entity.Position == v3 260.0f -260.0f 0.0f
    Entity.Elevation == 10.0f
    Entity.ClickEvent ==> msg Quit]]

    // the scene group
    Content.groupIfScreenSelected screen $ fun _ ->
        Content.group Simulants.Gameplay.Scene.Group.Name []
        [Content.character Simulants.Gameplay.Scene.Player.Name
        [Entity.Position == v3 0.0f 0.0f 0.0f
        Entity.Size == v3 108.0f 108.0f 0.0f]]

    // the level group
    Content.groupIfScreenSelected screen $ fun _ ->
        Content.groupFromFile Simulants.Gameplay.Level.Group.Name "Assets/Gameplay/Level.nugroup"]

```


Finally, let's look at how the simulant's themselves are structured in **MySimulants.fs** -

```
namespace MyGame
open Nu

// this module provides global handles to the game's key simulant's.
// having a Simulant's module for your game is optional, but can be nice to avoid duplicating string literals across
// the code base.
[<RequireQualifiedAccess>]
module Simulant's =

    [<RequireQualifiedAccess>]
    module Splash =

        let Screen = Screen "Splash"

    [<RequireQualifiedAccess>]
    module Title =

        let Screen = Screen "Title"

    [<RequireQualifiedAccess>]
    module Gui =

        let Group = Screen / "Gui"
        let Play = Group / "Play"
        let Credits = Group / "Credits"
        let Exit = Group / "Exit"

    [<RequireQualifiedAccess>]
    module Credits =

        let Screen = Screen "Credits"

    [<RequireQualifiedAccess>]
    module Gui =

        let Group = Screen / "Gui"
        let Back = Group / "Back"

    [<RequireQualifiedAccess>]
    module Gameplay =

        let Screen = Screen "Gameplay"

    [<RequireQualifiedAccess>]
    module Gui =

        let Group = Screen / "Gui"
        let Quit = Group / "Quit"

    [<RequireQualifiedAccess>]
    module Scene =

        let Group = Screen / "Scene"
        let Player = Group / "Player"
```

```
[<RequireQualifiedAccess>]
module Level =


    let Group = Screen / "Level"
```

And that's all there is to it!

Before discussing Nu's game engine design and how to customize your game, let's have a little fun messing around with Nu's real-time interactive editor, **Gaia**.

What is Gaia?

Gaia is Nu's game editing tool. Here is a screenshot of an empty editing session –




Run Gaia by setting the **Gaia project** as the **StartUp Project** in Visual Studio, and then running.

Upon starting, you'll notice the **Editor Start Configuration** screen. If you select your game's .NET executable, the custom types that you exposed in your **Plugin** type will be available for use in the editor. If you cancel this dialog, you get only what comes with Nu out of the box. Additionally, you can have the editor open your gameplay screen (assuming you have it assigned to **Default.Screen** as in the above template game).


Here we will just cancel the dialog and play with the dispatchers / facets that come out-of-the-box.

First, let's create a blank button in Gaia by selecting **ButtonDispatcher** from the combo box to the right of the **Create Entity** button, and then pressing the **Create Entity** button -



We have a button! Now notice that the property grid on the right has been populated with its properties. These properties can be edited to change the button however you like. For a button that will be used to control the game's stat, the first thing you will want to do is to give it an appropriate name. Do so by double-clicking the **Name** property, deleting the contents, and then entering the text **MyButton**. Naming entities give you the ability to access them at run-time via that name once you have loaded the containing document in your game.


Notice also that you can click and drag on the button to move it about the screen. You can also right-click and entity for additional operations via a context menu. Let's change the **Text** property of the button to "Hello World!" and move the button to the bottom right of the screen -



Let's now try putting Gaia in **Ticking** mode so that we can test that our button clicks as we expect. Toggle on the **Ticking** button at the top right, then click on the button.

Once you're satisfied, toggle off the **Ticking** button to return to the non-ticking mode.


Now let's make a default tile map to play around with. BUT FIRST, we need to change the depth of our button entity so that it doesn't get covered by the new tile map. Change the value in the button's **Depth** property to **10**.



drop down box to the right of the **Create Entity** button, select (or type) **TileMapDispatcher**, and then press the **Create Entity** button. You'll get this –


Let's rename the tile map to **MyTileMap**. Then, let us click and drag the tile map so its bottom-left corner lines up with the bottom left of the editing panel.

Tile maps, by the way, are created with the free tile map editor **Tiled** found at <http://www.mapeditor.org/>. All credit to the great chap who made and maintains it!



Now click and drag with the MIDDLE mouse button to change the position of the camera that is used to view the game. Check out your lovely new tile map! If your camera gets lost in space, click the **Reset Camera** button that is to the left of the **Ticking** button.

Now let's create some boxes that use physics to fall down and collide with the tile map. First, we must change the default depth at which new entities are created (again, so the tile map doesn't overlap them). In the **at Depth** text box to the left of the **Quick Size** button, type in a **1** or click the **+** button to its right. In the combo box to the right of the **Create Entity** button, select (or type) **BoxDispatcher**, and then click the **Create Entity** button. You'll see a box that was created in the middle of the screen.



The **BoxDispatcher** is affected by gravity, so try moving the box upward and letting it fall. Why does it not fall? Well, the physics system is not enabled unless the game is ticking. But according to the **Ticking** toggle button at the top left, ticking is not toggled on. So let us toggle it on, and watch the box fall according to gravity!

Turn off **Ticking** when you're satisfied.


Another way to create boxes is by right-clicking at the desired location and then, in the context menu that pops up, clicking **Create**.

We can now save the document for loading into a game by clicking **File -> Save...**


Shut down the editor when you're done.

Loading Your Game in Gaia

Next, let's load up our newly-created game in the editor! A nice feature of Nu is that you can play your game directly in the editor while editing. Before starting, make sure your game is built. If you want to make sure it's always up to date before running Gaia, add it as a **Build Dependency** to the **Nu.Gaia** project. Once we're ready start a new session of Gaia and fill out the start dialog like so -



You will see the game loaded in the editor like so -



By selecting the tab of a containing Group at the top left, you can edit that Group's child simulants. So if you want to select and edit the Back button, have the **Hud** tab selected. To edit the Player character, select the **Scene** tab.

The Game Engine

By looking at the initial example, you might be able to make a vague inference of how Nu is used and structured. Let's try to give you a clearer idea.

First and foremost, Nu was designed for **games**. This may seem an obvious statement, but it has some implications that vary it from other middleware technologies.

Nu comes with an appropriate game structure out of the box, allowing you to house your game's implementation inside of it. Here's the overall structure of a game as prescribed by Nu –

World --> Game --> [Screen] --> [Group] --> [Entity]

In the above diagram, $X \rightarrow Y$ denotes a one-to-many relationship, and $[X] \rightarrow [Y]$ denotes that each X has a one-to-many relationship with Y . So for example, there is only one **Game** in existence, but it can contain many **Screens** (such as a 'Title Screen' and a 'Credits Screen'). Each **Screen** may contain multiple **Groups** that may in turn each contain multiple **Entities**.

Everyone should know by now that Gui (*graphical user interface*) elements are an intrinsic part of games. Rather than tacking on a Gui system like other engines, Nu implements its Gui components directly as entities. There is no arbitrary divide between a box with physics and a Gui button – they are both built from the same abstractions.

Let's break down what each of Nu's most important types mean in detail.

World

We already know a bit about the World type. As you can see in the above diagram, it contains the simulation values starting with the Game. In addition to that, it contains facilities needed to execute a game such as various subsystems (such as a render context, an audio context, physics, and those defined by the user), a purely-functional event system (far more appropriate to a functional game than .NET's or even F#'s mutable event systems), additional state values beyond the simulants shown above, and other types of dependencies. When you want something in your game to change, you operate on a World value to produce another World value.

Screen

Screens are precisely what they sound like – a way to implement a single 'screen' of interaction in your game. In Nu's conceptual model, a game is nothing more than a series of interactive screens to be traversed like a graph. The main simulation occurs within a given screen, just like everything else.

Group

Groups represent logical collections of entities that can be combined to make up a Screen. Each group has a Visible property that can be used to hide or show all of its entities.

Entity

And here we come down to brass tacks. Entities represent individual interactive 'things' in your game. We've seen several already – a button, a tile map, and boxes. What differentiates a button entity from a box entity, though? Each entity picks up its unique attributes from its **dispatcher**. What is a dispatcher? Well, it's a little complicated, so we'll touch on that slightly later! Please be patient ☺

Game Engine Details

Simulant Handles

Simulants are not accessed and transformed directly, but rather through handle types such as **Entity**, **Group**, **Screen**, and **Game**. Simulant handles are created from addresses that uniquely identify a given simulant - EG, **let entity = Entity entityAddress**. We'll elaborate more on addresses next.

Addresses

You may be wondering how the engine locates specific entities as created in Gaia and loaded from the saved ***.nulyr** file. All entities, and other simulants, are located by constructing an **address** that uniquely identifies where it exists in an internal map in the engine. Each entity has an address of the form **ScreenName/GroupName/EntityName**, where **ScreenName** is the name that is given to its containing screen, **GroupName** is the name given to its containing group, and **EntityName** is the name given to the entity (such as in the editor). Remember how we changed the **Name** property of the button object that we created to **MyButton** earlier in **Gaia**? That's the **EntityName** portion of its address! The same structure applies to screen and group addresses, albeit with fewer names. Game addresses are actually empty since there is only ever one game per world, thus no unique identifying information is needed.

Notice that addresses have a single type parameter that is used to make their intended usage more explicit. Addresses are used to both identify simulants as well as specify the events that take place upon them. You can tell the difference between simulant and event addresses by their type arguments, and even among different simulant and event types! Addresses used to locate simulants are typed according to the type of simulant they locate, and addresses that are used to specify events are typed according to the type of data their event carries.

For example, the **Events.MouseMove** has a generic type of **MouseMoveData**, and an **EntityAddress** has a generic type of **Entity**. Additionally, there are several operators and conversion functions used to combine addresses and manipulate their type appropriately in the **Address.fs** and **SimulationOperators.fs** files of the **Nu** project. With these functions, you can combine simulant addresses with the common event address value found in **SimulationEvents.fs** to specify event addresses as needed. This may all initially seem a little complicated, but please trust that this extra specificity it will save you from innumerable runtime errors.

The Functional Event System

Because the event system that F# provides out of the box is inherently mutating / impure, I had to invent a custom, purely-functional event system for Nu.

Subscriptions are created by invoking the **World.subscribe** function, and destroyed using the **World.unsubscribe** function. Since subscriptions are to address rather than particular simulants, you can subscribe to any address regardless of whether there exists a simulant there or not!

Additionally, there is a function that subscribes to events only for the lifetime of the subscriber. It is **World.monitor**. You will likely be using this more often than the other two functions as it more compactly provides the desired behavior.

I won't cover this in too much detail since, for the most part, you'll be using the Elm-style bindings to wire up your events for you.

Xtensions

Xtensions are a key enabling technology in Nu. Xtensions allow the **Game**, **Screen**, **Group**, and **Entity** types to be extended by the end-user in a purely-functional way. This extensibility mechanism is the key creating your own simulation types.

Xtensions Under the Hood

Perhaps the most efficient way to exemplify the usage of an Xtension is by discussing its unit tests. Be aware that in the following tests Xtensions are exercised in isolation, though of course the engine uses them by embedding them in a type as above. Let's take a look a snippet from Prime's Tests.fs file –

```
let [<Fact>] canAddProperty () =  
    let xtn = Xtension.empty  
    let xtn = xtn?TestProperty <- 5  
    let propertyValue = xtn?TestProperty  
    Assert.Equal (5, propertyValue)
```

For the first test, you can see we're using the Xtension type directly rather than embedding it in another type. This is not the intended usage pattern, but it does simplify things in the context of this unit test. The test here merely demonstrates that a property called **TestProperty** with a value of 5 can be added to an Xtension **xtn**.

At the beginning of the test, **xtn** starts out life as an Xtension value with no properties (the 'empty' Xtension). By using the **dynamic (<-)** operator as shown on the third line, **xtn** is augmented with a property named **TestProperty** that has a value of 5. The next line then utilizes the **dynamic (?)** operator to retrieve the value of the newly added property into the **propertyValue** variable. Note the surprising presence of strong typing on the **propertyValue** variable. Let's get an explanation of why we capture such strong typing here, and where capturing the typing otherwise would require a type annotation. Consider the following where type information isn't captured –

```
let typeInfoExample () =  
    let xtn = Xtension.empty  
    let xtn = xtn?TestProperty <- 5  
    let propertyValue = xtn?TestProperty  
    propertyValue
```

The type of this function will be **'a**. This is likely not what we want since we know that the returned value is intended to be of type **int**. To address this shortcoming, a type annotation is required. There are multiple ways to achieve this, but in order to maximize clarity, I suggest putting the type annotation as near as possible to its target like so –

```
let typeInfoExample () =  
    let xtn = Xtension.empty  
    let xtn = xtn?TestProperty <- 5  
    let propertyValue = xtn?TestProperty : int  
    propertyValue
```

An **int** annotation was added to the end of the fourth line, and the function's type became **unit -> int**. This is the level of type information we typically want and expect from F# code.

How Nu uses Xtensions in practice

Having seen the use of Xtensions in the narrow context of its unit tests, we need to understand how they're actually used in Nu.

First, note that the Xtension's properties are not usually accessed directly, but only accessed through each containing types' forwarding functions (as seen in the above Entity type definition). Further, in order to preserve the most stringent level of typing, user code doesn't use even the forwarding operators directly, but rather type extension functions like these –

```
type Entity with  
  
    member this.GetDensity world = this.Get<single> Property? Density world  
    member this.SetDensity value world = this.Set<single> Property? Density value world  
    member this.Density = Lens.make Property? Density this.GetDensity this.SetDensity this
```

- which, when used in practice, looks like this –

```
let world = entity.SetDensity 1.0f world
```

This is to allow user code to use the most stringent level of typing possible even though such properties are, in actuality, dynamic!

You'll also notice the member **Density** of type **Lens**. Each property should be accompanied by a related lens in order for it to participate in Nu's Elmish / MVU programming model. The lens is used to specify the initial value of the property, construct change events, among other things.

Dispatchers

A **dispatcher** is a stateless object that allows you to specify the behavior of a simulation type. Dispatchers are a simple implementation of a technique that harkens back to the **Strategy Pattern** of OOP, but are totally stateless. So they're not really objects in the object-oriented sense, but rather a convenient way that Nu's borrows dispatch polymorphism from .NET's object-oriented constructs. Overriding a dispatcher's methods is how we hook our simulant's custom behavior into the engine.

The **Register** method allows you to customize what happens to the simulant (and the world) when it is added to the world. **Unregister** allows you to customize what happens when it is removed. **Update** is your typical update callback, and **PostUpdate** is well, the post-update. **Actualize** is what can be implemented if you have some custom rendering that you want to implement.

All these overrides and more are available for you to customize your simulant's behavior. But that's not the only way. You can instead use the generic (such as **EntityDispatcher<_,_,>**) type to implement your entity using the Elm-style with its available overrides -

```
abstract member Prepare : 'model * World -> 'model
default this.Prepare (model, _) = model

abstract member Channel : Lens<'model, World> * Entity -> Channel<'message, 'command, Entity, World> list
default this.Channel (_, _) = []

abstract member Initializers : Lens<'model, World> * Entity -> PropertyInitializer list
default this.Initializers (_, _) = []

abstract member Physics : Vector2 * single * Vector2 * single * 'model * Entity * World -> Signal<'message, 'command> list * 'model
default this.Physics (_, _, _, _, model, _, _) = just model

abstract member Message : 'model * 'message * Entity * World -> Signal<'message, 'command> list * 'model
default this.Message (model, _, _, _) = just model

abstract member Command : 'model * 'command * Entity * World -> Signal<'message, 'command> list * World
default this.Command (_, _, _, world) = just world

abstract member Content : Lens<'model, World> * Entity -> EntityContent list
default this.Content (_, _) = []

abstract member View : 'model * Entity * World -> View
default this.View (_, _, _) = View.empty
```

Generally, the Elm-style of implementing entity is recommended unless you have some reason to use the lower-level style.

Facets

Don't we need some form a composition in order to reuse behaviors among different entities?

Of course, and that's what **Facets** are for!

A **Facet** implements a single, composable behavior that can be assigned to an entity. Like a dispatcher, a facet is a complete stateless object with override-able methods. Many of its method match the shape of an EntityDispatcher's as well. Let's look at the definition and use of one of Nu's most basic facets now –

[<AutoOpen>]

```
module StaticSpriteFacetModule =

    type Entity with
        member this.GetStaticImage world : Image AssetTag = this.Get Property? StaticImage world
        member this.SetStaticImage (value : Image AssetTag) world = this.Set Property? StaticImage value world
        member this.StaticImage = lens Property? StaticImage this.GetStaticImage this.SetStaticImage this
        member this.GetInsetOpt world : Vector4 option = this.Get Property? Inset world
        member this.SetInsetOpt (value : Vector4 option) world = this.Set Property? Inset value world
        member this.InsetOpt = lens Property? Inset this.GetInsetOpt this.SetInsetOpt this
        member this.GetBlend world : Blend = this.Get Property? Blend world
        member this.SetBlend (value : Blend) world = this.Set Property? Blend value world
        member this.Blend = lens Property? Blend this.GetBlend this.SetBlend this
        member this.GetFlip world : Flip = this.Get Property? Flip world
        member this.SetFlip (value : Flip) world = this.Set Property? Flip value world
        member this.Flip = lens Property? Flip this.GetFlip this.SetFlip this

    type StaticSpriteFacet () =
        inherit Facet ()

        static member Properties =
            [define Entity.StaticImage Assets.Default.Image4
            define Entity.Color Color.White
            define Entity.Blend Transparent
            define Entity.Glow Color.Zero
            define Entity.InsetOpt None
            define Entity.Flip FlipNone]

        override this.Actualize (entity, world) =
            if entity.GetVisible world && entity.GetInView world then
                let transform = entity.GetTransform world
                let staticImage = entity.GetStaticImage world
                World.enqueueRenderLayeredMessage
                { Elevation = transform.Elevation
                  PositionY = transform.Position.Y
                  AssetTag = AssetTag.generalize staticImage
                  RenderDescriptor =
                      SpriteDescriptor
                      { Transform = transform
                        Absolute = entity.GetAbsolute world
                        Offset = Vector2.Zero
                        InsetOpt = entity.GetInsetOpt world
                        Image = staticImage
                        Color = entity.GetColor world
                        Blend = entity.GetBlend world
                        Glow = entity.GetGlow world
                        Flip = entity.GetFlip world }}
            world

        override this.GetQuickSize (entity, world) =
            match World.tryGetTextureSizeF (entity.GetStaticImage world) world with
            | Some size -> size
            | None -> Constants.Engine.EntitySizeDefault
```

As you may see, the **StaticSpriteFacet** is used to define simple, static sprite rendering behavior for an entity.

Complex behavior for an entity dispatcher can be defined by composing together multiple facets. Here's a dispatcher that combines the SpriteFacet and RigidBodyFacet facets at compile-time –


```
[<AutoOpen>]
module RigidSpriteDispatcherModule =

    type RigidSpriteDispatcher () =
        inherit EntityDispatcher ()


    static member Facets =
        [typeof<RigidBodyFacet>
         typeof<SpriteFacet>]
```

Additionally, facets can be dynamically added to a removed from an entity in Gaia simply by changing the **FacetNames** property. Let's take a look.

Here we just create a vanilla entity by selecting **EntityDispatcher** and pressing the **Create Entity** button –



Notice how nothing appears in the editing panel. This is because a plain old EntityDispatcher does not come with any rendering functionality. Let's add that now by changing its **FacetNames** property to **[StaticSpriteFacet]** –



Not only does it now render, the additional properties needed to specify how rendering is performed are provided in the property grid (to the right). Try adding physics to the entity by changing **FacetNames** to **[RigidBodyFacet StaticSpriteFacet]**, and then toggling on the **Ticking** button.

It falls away! By creating your own facets and assigning them either statically like the code above or dynamically in the editor, there's no end to the behavior you can compose!

Assets and the AssetGraph

Nu has a special system for efficiently and conveniently handling assets called the **Asset Graph**. The Asset Graph is configured in whole by a file named **AssetGraph.nuag**. This file is included in every new Nu game project, and is placed in the same folder as the project's **Program.fs** file.

The first thing you might notice about assets in Nu is that they are not built like normal assets via Visual Studio. The Visual Studio projects themselves need to have no knowledge of a game's assets. Instead, assets are built by a program called **Nu.Pipe.exe**. Nu.Pipe knows what assets to build by itself consulting the game's Asset Graph. During the build process of a given Nu game project, Nu.Pipe is invoked from the build command line like so –

```
"$(ProjectDir)..\..\Nu\Nu.Pipe\bin\$(ConfigurationName)\Nu.Pipe.exe" "$(ProjectDir)" "$(TargetDir)" "$(ProjectDir)refinement" False
```

Nu.Pipe references the game's Asset Graph to automatically copy all its asset files to the appropriate output directory. Note that for speed, Nu.Pipe copies only missing or recently altered assets.

Let's study the structure of the data found inside the AssetGraph.nuag file that ultimately defines a game's Asset Graph

```
[[Default
  [[Asset Font "Assets/Default/FreeMonoBold.032.ttf" [Render] []]
   [Assets Assets/Default nueffect [Symbol] []]
   [Assets Assets/Default nuscript [Symbol] []]
   [Assets Assets/Default csv [Symbol] []]
   [Assets Assets/Default png [Render] []]
   [Assets Assets/Default wav [Audio] []]
   [Assets Assets/Default ogg [Audio] []]
   [Assets Assets/Default tmx [] []]]]]
```

This file uses Nu's s-expression syntax. There is a single **Package** that holds multiple **Asset** descriptors. In Nu, a single asset will never be loaded by itself. Instead, a package of assets containing the desired asset is loaded (or unloaded) all at once. The Asset Graph allows you to conveniently group together related assets in a package so they can be loaded as a unit.

Further, the use of the Asset Graph allows (well, *forces*) you to refer to assets by their asset and package name rather than their raw file name. Instead of setting a sprite image property to **Assets/Default/Image.png** (which absolutely will not work), you must instead set it to **[Default Image]** (assuming you want to load it from the **Default** package).

You may notice that there is no need to manually specify which assets will be loaded in your game before using them. This is because when an asset is used by the render or audio system, it will automatically have its associated package loaded on-demand. This is convenient and works great in Gaia, but this is not always what you want during gameplay. For example, if the use of an asset triggers a package load in the middle of an action sequence, the game could very well stall during the IO operations, thus resulting in an undesirable pause. Whenever this happens, a note will be issued to the console that an asset package was loaded on-the-fly. Consider this a performance warning for your game.

Fortunately, there is a simple way to alleviate the potential issue. When you know that the next section of your game will require a package of rendering assets, you can send a 'package use hint' to the renderer like so –

```
let world = World.hintRenderPackageUse "MyPackageName" world
```

Currently, this will cause the renderer to immediately load all the all the assets in the package named **MyPackageName** which are associated with the render system (which assets are associated with which system(s) is specified by the **associations** attribute of the Asset node in AssetGraph.nuag). Notice that this message is just a hint to the renderer, not an overt command. A future renderer may have different underlying behavior such as using asset streaming.

Conversely, when you know that the assets in a loaded package won't be used for a while, you can send a 'package disuse hint' to unload them via the corresponding **World.hintRenderPackageDisuse** function.

Finally, there is a nifty feature in Gaia where the game's currently loaded assets can be rebuilt and reloaded at run-time. This is done by pressing the **Reload Assets** button found at the top-right of the window.

Serialization

By default, all of your simulation types can be serialized at any time to a file. No extra work will generally be required on your behalf to make serialization work, even when making your own custom dispatchers.

To manually stop any given property from being serialized, you can define it with the **nonPersistent** function rather than the **define** function in the type's **Properties** list like in the following snippet -


```
static member Properties =
    [define Entity.AnimationDelay 3L
      define Entity.AnimationSheet Assets.Gameplay.PlayerImage
      nonPersistent Entity.LastTimeOnGround Int64.MinValue
      nonPersistent Entity.LastTimeJump Int64.MinValue]
```

Overlays

Overlays accomplish two extremely important functions in Nu. First, they reduce the amount of stuff written out to (and consequently read in from) serialization files. Second, they provide the user with a way to abstract over property values that multiple entities hold in common. User-defined overlays, called ‘overlay routes’, are defined in a file that is included with every new Nu game project called **Overlayer.nuol**. Additionally, for every dispatcher and facet type that the engine is informed of, an overlay route with a matching name is defined with values set to the type’s **Properties**.

Let’s look at the OmniBlade’s overlay definitions –

```
[[OmniButtonOverlay
  [ButtonOverlay]
  [ButtonDispatcher]
  [[ClickSoundOpt [Some [Gui Affirm]]]
   [DownImage [Gui ButtonDown]]
   [Font [Gui Font]]
   [TextColor [1 1 1 1]]
   [TextDisabledColor [.75 .75 .75 .75]]
   [UpImage [Gui ButtonUp]]]]]
[OmniTextOverlay
  [TextOverlay]
  [TextDispatcher]
  [[Font [Gui Font]]
   [TextColor [1 1 1 1]]
   [TextDisabledColor [.75 .75 .75 .75]]]]]
[OmniToggleOverlay
  [ToggleOverlay]
  [ToggleDispatcher]
  [[ToggleSoundOpt [Some [Gui Affirm]]]
   [Font [Gui Font]]
   [TextColor [1 1 1 1]]
   [TextDisabledColor [.75 .75 .75 .75]]]]]
```

Let’s understand the **ButtonOverlay** in isolation -

```
[OmniButtonOverlay
  [ButtonOverlay]
  [ButtonDispatcher]
  [[ClickSoundOpt [Some [Gui Affirm]]]
   [DownImage [Gui ButtonDown]]
   [Font [Gui Font]]
   [TextColor [1 1 1 1]]
   [TextDisabledColor [.75 .75 .75 .75]]
   [UpImage [Gui ButtonUp]]]]]
```

Obviously enough, the first field, **OmniButtonOverlay**, is the name of the overlay.

The second field, `[ButtonOverlay]`, is the list of overlays that the overlay will inherit properties from. This list is composed of a single ‘intrinsic’ overlay composed of the default properties of the ButtonDispatcher. It’s called an ‘intrinsic’ overlay because it is automatically created by the engine.

The third field, `[ButtonDispatcher]`, is the list of dispatchers that the overlay will be automatically applied to.

The remaining list is the set of properties that will be overlaid and their overlaid values.

Overlays avoid a ton of duplication while allowing changes to them to automatically propagate to the entities to which they are applied.

Subsystems and Message Queues

Fortunately, Nu is not a monolithic game engine. The definition of its simulation types and the implementation of the subsystems that process / render / play them are separate. They are so separate, in fact, that neither the engine, nor the dispatchers that define the behavior of simulation types, are allowed to send commands to the subsystems directly (note I said 'commands', the engine does send non-mutating queries the subsystems directly, though user code should not even do this). Instead of sending commands directly, each subsystem must be interfaced with via its own respective message queue.

Thankfully, there are convenience functions on the World type that make this easy. Remember the **World.hintRenderPackageUse** function? That is one of these convenience functions, and all of them are as easy to use. However, accessing additional functionality from any of the subsystems will require writing new messages for them, in turn requiring a change to the engine code. Fortunately, there is an easy way to enable creating new types of messages without requiring changes to the engine, and that will be implemented shortly (if it hasn't already by the time you read this).


Now, of course the use of message queues can make accomplishing certain things a little more complicated due to the inherent indirection it entails. Not only is the call-site a bit separated from the target, the time at which the actual message is handled is also separated. These two facts can make debugging a little more challenging. What does this indirection buy us that such additional difficulty is warranted?

For one, you'll notice that the API presented by each of the subsystems is inherently impure / stateful. If either the engine or user code were to invoke these APIs directly, the functional purity of both would be compromised, and all the nice properties that come from it destroyed. And secondly, it is likely that one or more of the subsystems will eventually be put on a thread separate from the game engine anyway, thus making the message queues unavoidable anyhow.

Currently, the subsystems used in Nu include a **Render** subsystem, an **Audio** subsystem, and a **Physics** subsystem. Additional subsystems such as **Artificial Intelligence** or a high-performance **Particle System** can be added by overriding the **MakeSubsystems** method in your **NuPlugin** subtype.

Special Effects System

A recent addition to the Nu Game Engine is the special effects system called **EffectSystem**.



To apply an effect to an Entity, you add the **EffectFacet** to its **FacetNames** property, and modify its **Effect** property to specify the desired effect using the effect syntax described below.

Effect Syntax

It is composed of a DSL using symbolic expressions, and a short list of composable semantics –

[Expand *definitionName* [*args...*]] – Expand a Content definition (more on definitions later).

[StaticSprite *resource* [*aspects...*] *childContent*] – Display a static sprite with the given Aspects (more on aspects later), and optional mounted Content.

[AnimatedSprite *resource* *celSize* *celRun* *celCount* *stutter* [*aspects...*] *childContent*] – Display an animated sprite with the given properties, Aspects, and optional mounted child Content.

[TextSprite *resource* *text* [*aspects...*] *childContent*] – Display a text sprite with the given text, Aspects, and optional mounted Content.

[Mount [Shift *shift*] [*aspects...*] *childContent*] – Mount the given child content with the given depth shift amount and Aspects.

[Repeat [Shift *shift*] ([Iterate *iterations*] | [Cycle *cycles*]) [*incrementAspects*] *childContent*] – Repeatedly invoke the given child Content with the given number of iterations or cycles of the given increment Aspects. Intuitively, like a declarative ‘for’ loop with either the incrementAspects applied iteratively or in a cycle.

[Emit [Shift *shift*] [Rate *rate*] [*emitterAspects...*] [*aspects...*] *childContent*] – Emit the given child Content at the given rate with the given Aspects. Note that due to performance limitations, Emit does not inherit its aspects from its parent, but has its own emitter Aspects.

Note also that emitters don't yet stack on other emitters due to a lack of implementation. This is coming soon, however!

[Composite [Shift *shift*] [*childContents...*]] – Compose multiple child Contents.

[Tag *name* `quotedValue`] – Tags an effect with given name paired with the given quoted value. Tags can be pulled from an Entity's EffectTags map at run-time via the given name, and have the quoted values observed or evaluated separately.

Nil – Specifies a lack of further Content.

As you might notice, the DSL syntax is built purely out of Nu's symbolic serialization syntax (or, symbolic-expressions (or *s-exprs* for short)). So to understand how the syntax operates in full, you simply need to reference the structure of the effect data types in the 'Nu/Nu/Nu/Effect.fs' source file.

Effect Aspects

In addition to normal parameters, Effect semantics allow modification via **Aspects**. Aspects specify the **Position**, **Size**, **Color**, and other properties of an Effect which can be animated over multiple **Key Frames** and inherited via implicit or explicit **Mounting**.

First, let's cover the general syntax of Aspects –

[AspectName *logic algorithm playback* [*keyFrames...*]] – The AspectName is, well, the name of the Aspect that is to be modified. Possibilities here include -

The logic value is any one of the following: **Or** | **Nor** | **Xor** | **And** | **Nand** | **Equal**. The logic value applies the value from the Aspect's current animation frame to its inherited property. So if the Enabled aspect is True on the current frame and the logic value is set to **And**, then the resulting Enabled property will be True if and only if the inherited Enabled value is True.

The remaining possible logic values are self-explanatory, except maybe for **Eq**. Eq simply takes the current frame value and ignores the inherited value.

The algorithm value exists on non-boolean Aspects. It is any one of the following: **Constant** | **Linear** | **Random** | **Chaos** | **Ease** | **EaseIn** | **EaseOut** | **Sin** | **SinScaled** | **Cos** | **CosScaled**.

The playback value is any one of the following: **Once** | **Loop** | **Bounce**. Once means that the animation will play once and then stop at the last Key Frame. Loop means the animation will play repeatedly from start to finish. Bounce means the animation will be played alternatively forward and backward.

Finally, of note, are the key frames, which have their own syntax like the following –

[[True 0] [False 10] [True 0]]

As you can see, it is just a list of Boolean values with the number of frames for which the value should hold. For the above, the value will be True on the first frame, False for 10 frames thereafter, then True after that. You can have as many Key Frames as you like.

Here are some example Key Frames for the Position aspect –

[[[0 0 0] 10] [[100 100 0] 170]]

It's a little more verbose since Vector2s need to be in their own list.

Here is the syntax for each Aspect in detail -

[Enabled *logic playback* [keyFrames...]] This property dictates whether an effect is enabled – which has a different meaning depending on the Content. For example, if the Content is StaticSprite, it dictates if the StaticSprite is displayed that frame. If a SoundEffect, it dictates if the sound effect is to be played that frame. If an Emit effect, it dictates if the emitter is emitting that frame.

TODO: rest of aspects!!!

Proper Effect Rendering with Entity Overflow

If you pan the screen off of an entity that's outputting an effect outside of its normal bounds, you may see the effect disappear unexpectedly. This is because Nu's culling system thinks the effect need not be processed due to the entity being out of culling bounds.

Therefore, it is important to understand the **Overflow** property of entities, and to adjust them properly.

The **Overflow** property of each entity expands the bounds of each entity by a multiple of its value. So if its value is [0 0] (which is the default), no bounds expansion happens. If it is [1 1], then the bounds are expanded by 100% of the original size, and so on.

So you must estimate the appropriate **Overflow** for entities with effects in order to adjust their bounds for proper culling.

Sample Effects

TODO

Particle System

TODO